

8 Device Attendance Commands

Introduction

PJL enables you to change messages displayed on the printer control panel. This capability is useful for alerting operators to perform specific actions, such as loading paper or pressing keys. This chapter describes the following three device attendance commands:

- RDYMSG specifies a “ready message” that replaces the “00 READY” message on the printer control panel. The RDYMSG command does not affect the online state.
- OPMSG displays a specified “operator message” on the printer control panel and takes the printer offline.
- STMSG displays the specified “status message” on the printer control panel and takes the printer offline. It returns the name of the key that the operator pressed to put the printer back online.

Note

Some PJL printers do not support the device attendance commands. See Appendix A for a list of features supported by each printer.

RDYMSG Command

The RDYMSG command replaces the control panel's "ready" message with the specified message. Use the RDYMSG command to modify the displayed message while keeping the printer online.

Syntax:

```
@PJL RDYMSG DISPLAY = "message" [<CR>]<LF>
```

Parameters:

Parameter	Functional Range	Default
DISPLAY = " <i>message</i> "	Roman-8 characters 33 and 35 thru 255, <SP>	N/A

- **DISPLAY = "*message*"** — The variable *message* can be any combination of printable characters (except quotation marks, character 34) and spaces, with a limit of 1 line of 16 characters (see "Comments" below). The *message* variable is a string and must be enclosed in double quotes as shown in the command syntax.

Comments:

Since the RDYMSG command leaves the printer online, it can be used to display the name of the person printing the current job, or other pertinent information, while a job is printing. It also can be used in network environments to display the name of the printer.

Note

For the HP LaserJet 4000, 5000, 8000, 8100, Color LaserJet 4500, and 8500 printers (and newer), use the `JOB DISPLAY="display text"` method to display a message on the front panel during a specific job. For these printers, RDYMSG only replaces the lowest priority message, which is typically only the READY message.

The command `@PJL RDYMSG DISPLAY = ""` (or performing a Menu Reset—"09 Menu Reset") returns the display to the normal ready message.

Note

This command replaces the lowest priority 00 READY message. If any higher priority message is displayed (for example, toner low), the new ready message does not appear until these outstanding conditions are cleared.

For HP LaserJet 4PJ, 4V, 4MV, 5Si, 5SiMx, and 5Si Mopier printers, if the LANG variable is set to Japanese, the message is displayed using the JIS X0201-76 character set.

For the LaserJet 4000, 5000, 8000, and 8100 series printers, if LANG is set to POLISH, CZECH, or HUNGARIAN, messages are displayed using the Latin 2 (ISO 8859/2) character set. If LANG is RUSSIAN, messages are displayed using Cyrillic (ISO 8859/5). For LANG=TURKISH, the display is in Latin 5 (ISO 8859/9). This is also true for the HP Color LaserJet 4500 printer, except there is no support for Turkish or Hungarian on this printer.

Some HP LaserJet printers may have a different limit for the number of display lines and number of display characters. To request the display limits, use the INFO CONFIG command. When this command is used, the printer returns DISPLAY LINES=*value* and DISPLAY CHARACTER SIZE=*value* as part of the response message. For example, HP LaserJet 4 and 4M printers return DISPLAY LINES=1, DISPLAY CHARACTER SIZE=16, indicating that the displayed message is limited to 1 line of 16 characters.

Example: Using the RDYMSG Command

This example simulates a spooler that prints a job from another application. The spooler uses the RDYMSG command to display the name of the person printing the job (while the job is printing).

Note

In the example below, the @PJL JOB DISPLAY command has been added to display the job name on the newer printers. (See the note on page 8-2.)

After the job is printed, the spooler restores the normal ready message. (The indented command lines indicate those sent by the printing application; the remaining lines are sent by the spooler.)

```
<ESC>%-12345X@PJL <CR><LF>
@PJL JOB NAME = "Terry's Job" <CR><LF>
@PJL JOB DISPLAY = "Terry's Job" <CR><LF>
@PJL RDYMSG DISPLAY = "TERRY'S JOB" <CR><LF>
  <ESC>%-12345X@PJL <CR><LF>
    @PJL ENTER LANGUAGE = PCL <CR><LF>
    <ESC>E . . . PCL Job . . .<ESC>E
    ↵<ESC>%-12345X
  ↵<ESC>%-12345X@PJL <CR><LF>
@PJL COMMENT Restore READY message <CR><LF>
@PJL RDYMSG DISPLAY = "" <CR><LF>
@PJL EOJ NAME = "End of Terry's Job" <CR><LF>
<ESC>%-12345X
```

Related Commands:

OPMSG, STMSG, INFO CONFIG

OPMSG Command

The OPMSG command prompts the printer to display the specified message and go offline. Use this command to display a message and halt printing until the operator presses the On Line, Continue, or Reset key.

Syntax:

```
@PJL OPMSG DISPLAY = "message" [<CR>]<LF>
```

Parameters:

Parameter	Functional Range	Default
DISPLAY = " <i>message</i> "	Roman-8 characters 33 and 35 – 255, <SP>,<HT>	N/A

- **DISPLAY = "*message*"** — The variable *message* can be any combination of printable characters (except quotation marks, character 34) and spaces or horizontal tab characters, with a limit of 1 line of 16 characters (see “Comments” below). The *message* variable is a string and must be enclosed in quotation marks as shown in the command syntax.

Comments:

On some HP printers, the first two display characters blink. To avoid blinking, begin the message with two blank spaces.

Some HP LaserJet printers may have a different limit for the number of display lines and number of display characters. To request the display limitations, use the INFO CONFIG command. The printer returns DISPLAY LINES=*value* and DISPLAY CHARACTER SIZE=*value* as part of the response message. For example, the HP LaserJet 4 and 4M printers return DISPLAY LINES=1, DISPLAY CHARACTER SIZE=16, indicating that the displayed message is limited to 1 line of 16 characters.

Note

For HP LaserJet 4PJ, 4V, 4MV, 5Si, 5SiMx, 5Si Mopier printers, if the LANG variable is set to Japanese, the message is displayed in the JIS X0201-76 character set.

For the LaserJet 4000, 5000, 8000, and 8100 series printers, if LANG is set to POLISH, CZECH, or HUNGARIAN, messages are displayed using the Latin 2 (ISO 8859/2) character set. If LANG is RUSSIAN, messages are displayed using Cyrillic (ISO 8859/51). For LANG=TURKISH, the display is in Latin 5 (ISO 8859/9). This is also true for the HP Color LaserJet 4500 printer, except there is no support for Turkish or Hungarian on this printer.

Example: Using the OPMSG Command

This example displays "LOAD LETTERHEAD" on the printer control panel and halts processing until the operator presses the On Line or Continue key. Once one of the keys is pressed, the first page is printed. Then "LOAD PLAIN PAPER" is displayed. When the On Line or Continue key is pressed, the printer prints the rest of the job.

```
<ESC>%-12345X@PJL <CR><LF>
@PJL JOB NAME = "JOB 111753" <CR><LF>
@PJL OPMSG DISPLAY = "LOAD LETTERHEAD" <CR><LF>
@PJL <CR><LF>
@PJL ENTER LANGUAGE = PCL <CR><LF>
<ESC>E This is "Letterhead"(page 1)<ESC>E
↵<ESC>%-12345X@PJL <CR><LF>
@PJL EOJ NAME = "EOJ 111753" <CR><LF>
<ESC>%-12345X
```

Related Commands:

RDYMSG, STMSG, INFO CONFIG

STMSG Command

The STMSG command prompts the printer to display the specified message, go offline, and return a status message indicating which key the operator pressed to return the printer online. Use this command to display a customized message, to halt printing until the operator presses the On Line, Continue, or Reset key, and to take a different action depending on which key the user presses.

Syntax:

```
@PJL STMSG DISPLAY = "message" [<CR>]<LF>
```

Return Syntax:

```
@PJL STMSG DISPLAY="message" <CR><LF>  
key <CR><LF>  
<FF>
```

Parameters:

Parameter	Functional Range	Default
DISPLAY = "message"	Roman-8 characters 33 and 35 – 255, <SP>,<HT>	N/A
key	ONLINE, RESET, CONTINUE, or JOBCANCEL	N/A

- **DISPLAY = "message"** — The variable *message* can be any combination of printable characters (except quotation marks, character 34) and spaces or horizontal tab characters, with a limit of 1 line of 16 characters (see comments below). The *message* variable is a string and must be enclosed in double quotes as shown in the command syntax.
- **key** — This variable is returned with the status response after the operator presses one of three control panel keys. The *key* variable has a value of either ONLINE, RESET, JOBCANCEL or CONTINUE, depending on which key the operator presses to return the printer online.

Note

For HP LaserJet 4PJ, 4V, 4MV, 5Si, 5SiMx, 5Si Mopier printers, if the LANG variable is set to Japanese, messages are displayed using the JIS X0201-76 character set.

For the LaserJet 4000, 5000, 8000, and 8100 series printers, if LANG is set to POLISH, CZECH, or HUNGARIAN, messages are displayed using the Latin 2 (ISO 8859/2) character set. If LANG is RUSSIAN, messages are displayed using Cyrillic (ISO 8859/5). For LANG=TURKISH, the display is in Latin 5 (ISO 8859/9). This is also true for the HP Color LaserJet 4500 printer, except there is no support for Turkish or Hungarian on this printer.

Comments:

This command is useful for making a programming decision based on which key the operator presses. For example, the application can prompt the user to press Reset, and then wait until the Reset key is pressed.

Some HP LaserJet printers may have a different limit for the number of display lines and number of display characters. To request the display limits, use the INFO CONFIG command. The printer returns DISPLAY LINES=*value* and DISPLAY CHARACTER SIZE=*value* as part of the response message. For example, HP LaserJet 4 and 4M printers return DISPLAY LINES=1, DISPLAY CHARACTER SIZE=16, indicating that the displayed message is limited to 1 line of 16 characters.

Example: Using the STMSG Command

This example prompts the printer to display “LOAD FORM 1040” and goes offline. The printer returns the name of the key that is pressed to resume operation—in this case, the Continue key.

```
<ESC>%-12345X@PJL <CR><LF>  
@PJL <CR><LF>  
@PJL STMSG DISPLAY = "LOAD FORM 1040" <CR><LF>  
@PJL <CR><LF>  
@PJL ENTER LANGUAGE = PCL <CR><LF>  
<ESC>E . . PCL job . .<ESC>E<ESC>%-12345X
```

The returned message, if the operator presses the Continue key, is:

```
@PJL STMSG DISPLAY="LOAD FORM 1040"  
CONTINUE  
<FF>
```

Related Commands:

OPMSG, RDYMSG, INFO CONFIG

